


Examples of Modeling Count Outcomes via SAS PROC GLIMMIX and GENMOD

The data for this example come from a recent study of the effects of emotion regulation strategy (none=control, cognitive reappraisal, or suppression) on aggressive behavior in persons with or without a history of intimate partner violence (IPV). The analysis was planned as a 2x3 between-groups ANCOVA with factors for strategy (3) and IPV (2), with neutral condition behavior as a covariate.

SAS Syntax and Output for Data Manipulation:

```
DATA example; SET example; LENGTH ERconds $10 IPV;
    IF ERcond=1 THEN ERconds="1_None";
    ELSE IF ERcond=2 THEN ERconds="2_CogR";
    ELSE IF ERcond=3 THEN ERconds="3_Supp";
    LABEL IPV= "Intimate Partner Violence (0=N,1=Y)"
        Neutral= "Aggressive during Neutral Condition (0=N,1=Y)"
        ERconds= "Condition (1=None, 2=CogR, 3=Supp)"
        Aggr= "Aggressive Verbalizations Outcome";
RUN;
PROC UNIVARIATE DATA=example; VAR Aggr; HISTOGRAM Aggr; RUN;
```



Look how non-aggressive our sample is! That's great for them, but not so good if we expect to use ANOVA to analyze this outcome...

However, this is only the marginal distribution of Y. Maybe the residuals will look more normal?

Model Predicting Aggressive Verbalizations using Normal Distribution and Identity Link (ANCOVA, usually estimated with least squares, here with ML using MIXED)

$$E(Aggr_p) = \beta_0 + \beta_1 Neutral_p + \beta_2 IPV_p + \beta_3 NoneVsCog_p + \beta_4 NoneVsSupp_p + \beta_5 IPV_p * NoneVsCog_p + \beta_6 IPV_p * NoneVsSupp_p + e_p$$

```
TITLE1 "GLM of Aggressive Verbalizations";
PROC MIXED DATA=example NOINFO NOITPRINT NOCLPRINT COVTEST NAMELEN=100 METHOD=ML;
    CLASS ERconds; * CLASS statement creates contrasts for us;
    MODEL Aggr = Neutral IPV ERconds IPV*ERconds / SOLUTION;
```

Covariance Parameter Estimates				
Cov Parm	Estimate	Standard Error	Z	Pr > Z
Residual	11.1803	1.0541	10.61	<.0001

Fit Statistics

-2 Log Likelihood	1181.7
AIC (smaller is better)	1197.7
AICC (smaller is better)	1198.4
BIC (smaller is better)	1225.0

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
Neutral	1	218	2.56	0.1109
IPV	1	218	0.92	0.3391
ERconds	2	218	4.31	0.0145
IPV*ERconds	2	218	2.73	0.0676

$p = .0676$, seriously?

LSMEANS to get cell means:


```
LSMEANS ERconds / AT (IPV Neutral)=(0 0); * Cell means per group for IPV=0;
LSMEANS ERconds / AT (IPV Neutral)=(1 0); * Cell means per group for IPV=1;
```

RUN;

Least Squares Means

Effect	Condition (1=None, 2=CogR, 3=Supp)	Condition		Estimate	Standard Error	DF	t Value	Pr > t
		Neutral	IPV					
ERconds	1_None	0.00	0.00	2.5915	0.4659	218	5.56	<.0001
ERconds	2_CogR	0.00	0.00	0.8242	0.4636	218	1.78	0.0768
ERconds	3_Supp	0.00	0.00	2.3272	0.4553	218	5.11	<.0001
ERconds	1_None	0.00	1.00	2.8577	0.7392	218	3.87	0.0001
ERconds	2_CogR	0.00	1.00	0.01703	0.7523	218	0.02	0.9820
ERconds	3_Supp	0.00	1.00	4.2921	0.6904	218	6.22	<.0001

What about that whole non-normal residuals thing? Yep, still an issue... what to do instead?



Model Predicting Aggressive Verbalizations using Poisson Distribution and Log Link

$$\begin{aligned} \log(E(\text{Aggr}_p)) = & \beta_0 + \beta_1 \text{Neutral}_p + \beta_2 \text{IPV}_p + \beta_3 \text{NoneVsCog}_p + \beta_4 \text{NoneVsSupp}_p \\ & + \beta_5 \text{IPV}_p * \text{NoneVsCog}_p + \beta_6 \text{IPV}_p * \text{NoneVsSupp}_p \end{aligned}$$

```
TITLE1 "Poisson Model of Aggressive Verbalizations";
PROC GLIMMIX DATA=example NOCLPRINT NAMELEN=100;
  CLASS ERconds; * CLASS statement creates contrasts for us;
  MODEL Aggr = Neutral IPV ERconds IPV*ERconds / SOLUTION LINK=LOG DIST=POISSON;
  LSMEANS ERconds / ILINK AT (IPV Neutral)=(0 0); * Cell means per group for IPV=0;
  LSMEANS ERconds / ILINK AT (IPV Neutral)=(1 0); * Cell means per group for IPV=1;
RUN;
```

Fit Statistics

-2 Log Likelihood	1144.96
AIC (smaller is better)	1158.96
AICC (smaller is better)	1159.48
BIC (smaller is better)	1182.87
CAIC (smaller is better)	1189.87
HQIC (smaller is better)	1168.61
Pearson Chi-Square	1016.55
Pearson Chi-Square / DF	4.66

Type III Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
Neutral	1	218	12.05	0.0006
IPV	1	218	3.45	0.0645
ERconds	2	218	21.94	<.0001
IPV*ERconds	2	218	9.64	<.0001

Now that's more like it! 😊

ERconds Least Squares Means

Condition (1=None, Standard 2=CogR, 3=Supp)			Standard		DF	t Value	Pr > t	Mean	Error Mean
	Neutral	IPV	Estimate	Error					
1_None	0.00	0.00	0.9439	0.08560	218	11.03	<.0001	2.5700	0.2200
2_CogR	0.00	0.00	-0.1232	0.1437	218	-0.86	0.3922	0.8841	0.1271
3_Supp	0.00	0.00	0.8440	0.08883	218	9.50	<.0001	2.3257	0.2066
1_None	0.00	1.00	1.0313	0.1283	218	8.04	<.0001	2.8047	0.3598
2_CogR	0.00	1.00	-1.9560	0.5777	218	-3.39	0.0008	0.1414	0.08170
3_Supp	0.00	1.00	1.4217	0.09998	218	14.22	<.0001	4.1442	0.4143

Model Predicting Aggressive Verbalizations using Negative Binomial Distribution and Log Link

$$\begin{aligned} \log(E(\text{Aggr}_p)) = & \beta_0 + \beta_1 \text{Neutral}_p + \beta_2 \text{IPV}_p + \beta_3 \text{NoneVsCog}_p + \beta_4 \text{NoneVsSupp}_p \\ & + \beta_5 \text{IPV}_p * \text{NoneVsCog}_p + \beta_6 \text{IPV}_p * \text{NoneVsSupp}_p + e_p \end{aligned}$$

```
TITLE1 "Negative Binomial Model of Aggressive Verbalizations";
PROC GLIMMIX DATA=example NOCLPRINT NAMELEN=100;
  CLASS ERconds; * CLASS statement creates contrasts for us;
  MODEL Aggr = Neutral IPV ERconds IPV*ERconds / SOLUTION LINK=LOG DIST=NEGBIN;
  LSMEANS ERconds / ILINK AT (IPV Neutral)=(0 0); * Cell means per group for IPV=0;
  LSMEANS ERconds / ILINK AT (IPV Neutral)=(1 0); * Cell means per group for IPV=1;
RUN;
```

Fit Statistics	
-2 Log Likelihood	817.95
AIC (smaller is better)	833.95
AICC (smaller is better)	834.61
BIC (smaller is better)	861.28
CAIC (smaller is better)	869.28
HQIC (smaller is better)	844.98
Pearson Chi-Square	247.87
Pearson Chi-Square / DF	1.14

Poisson model $-2LL = 1144.96$

$-2\Delta LL(df = 1) = 1146.96 - 817.95 = 329.01, p = 1.57911E-73$

So the model fits significantly better from adding a “dispersion” parameter that allows the variance to exceed the mean.

Parameter Estimates						
Effect	Condition (1=None, 2=CogR, 3=Supp)	Estimate	Standard Error	DF	t Value	Pr > t
Neutral		0.4493	0.3634	218	1.24	0.2177
IPV		0.5671	0.3391	218	1.67	0.0959
ERconds	1_None	0.07835	0.2760	218	0.28	0.7768
ERconds	2_CogR	-0.9583	0.2963	218	-3.23	0.0014
ERconds	3_Supp	0
IPV*ERconds	1_None	-0.4379	0.4941	218	-0.89	0.3765
IPV*ERconds	2_CogR	-2.4299	0.7596	218	-3.20	0.0016
IPV*ERconds	3_Supp	0
Scale		1.5789	0.2408	.	.	.

Type III Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
Neutral	1	218	1.53	0.2177
IPV	1	218	1.91	0.1682
ERconds	2	218	7.26	0.0009
IPV*ERconds	2	218	5.12	0.0067

ERconds Least Squares Means									
Condition (1=None, Standard 2=CogR, 3=Supp)			Standard		DF	t Value	Pr > t	Mean	Error Mean
	Neutral	IPV	Estimate	Error					
1_None	0.00	0.00	0.9296	0.1983	218	4.69	<.0001	2.5334	0.5024
2_CogR	0.00	0.00	-0.1071	0.2254	218	-0.47	0.6353	0.8985	0.2025
3_Supp	0.00	0.00	0.8512	0.1927	218	4.42	<.0001	2.3425	0.4514
1_None	0.00	1.00	1.0588	0.3043	218	3.48	0.0006	2.8829	0.8772
2_CogR	0.00	1.00	-1.9699	0.6469	218	-3.05	0.0026	0.1395	0.09022
3_Supp	0.00	1.00	1.4183	0.2796	218	5.07	<.0001	4.1301	1.1547

ESTIMATE statements to further decompose the interaction (requesting all simple effects);

```
ESTIMATE "No/Yes IPV Difference: No Instruction"          IPV 1 IPV*ERconds 1 0 0;
ESTIMATE "No/Yes IPV Difference: Cognitive Reappraisal"  IPV 1 IPV*ERconds 0 1 0;
ESTIMATE "No/Yes IPV Difference: Suppression "          IPV 1 IPV*ERconds 0 0 1;
ESTIMATE "None/Cog Difference: No IPV"                  ERconds -1 1 0 IPV*ERconds 0 0 0;
ESTIMATE "None/Sup Difference: No IPV"                   ERconds -1 0 1 IPV*ERconds 0 0 0;
ESTIMATE "Cog/Supp Difference: No IPV"                   ERconds 0 -1 1 IPV*ERconds 0 0 0;
ESTIMATE "None/Cog Difference: Yes IPV"                  ERconds -1 1 0 IPV*ERconds -1 1 0;
ESTIMATE "None/Sup Difference: Yes IPV"                  ERconds -1 0 1 IPV*ERconds -1 0 1;
ESTIMATE "Cog/Supp Difference: Yes IPV"                  ERconds 0 -1 1 IPV*ERconds 0 -1 1;
ESTIMATE "No/Yes IPV by None/Cog Interaction"            IPV*ERconds -1 1 0;
ESTIMATE "No/Yes IPV by None/Sup Interaction"            IPV*ERconds -1 0 1;
ESTIMATE "No/Yes IPV by Cog/Sup Interaction"              IPV*ERconds 0 -1 1;
```

Label	Estimates		DF	t Value	Pr > t
	Estimate	Standard Error			
No/Yes IPV Difference: No Instruction	0.1292	0.3578	218	0.36	0.7184
No/Yes IPV Difference: Cognitive Reappraisal	-1.8628	0.6830	218	-2.73	0.0069
No/Yes IPV Difference: Suppression	0.5671	0.3391	218	1.67	0.0959
None/Cog Difference: No IPV	-1.0366	0.2966	218	-3.49	0.0006
None/Sup Difference: No IPV	-0.07835	0.2760	218	-0.28	0.7768
Cog/Supp Difference: No IPV	0.9583	0.2963	218	3.23	0.0014
None/Cog Difference: Yes IPV	-3.0286	0.7113	218	-4.26	<.0001
None/Sup Difference: Yes IPV	0.3595	0.4082	218	0.88	0.3794
Cog/Sup Difference: Yes IPV	3.3882	0.6998	218	4.84	<.0001
No/Yes IPV by None/Cog Interaction	-1.9920	0.7716	218	-2.58	0.0105
No/Yes IPV by None/Sup Interaction	0.4379	0.4941	218	0.89	0.3765
No/Yes IPV by Cog/Sup Interaction	2.4299	0.7596	218	3.20	0.0016

Also examined Zero-Inflated Poisson and Negative Binomial, just to be sure:

```
TITLE1 "ZIP Model of Aggressive Verbalizations";
PROC GENMOD DATA=example NAMELEN=100;
  CLASS ERconds; * CLASS statement creates contrasts for us;
  MODEL Aggr = Neutral IPV ERconds IPV*ERconds / LINK=LOG DIST=ZIP;
  ZEROMODEL / LINK=LOGIT; RUN;
```

Criteria For Assessing Goodness Of Fit	
Criterion	Value
Full Log Likelihood	-489.0974
AIC (smaller is better)	994.1948
AICC (smaller is better)	994.8615
BIC (smaller is better)	1021.5236

Negative Binomial $-2LL = 817.95$,
 AIC =833.95, BIC = 861.28

AIC and BIC are higher for ZIP, so NB is better.
 Below, the logit of being an “extra 0” = -0.5461 .

Analysis Of Maximum Likelihood Zero Inflation Parameter Estimates

Parameter	DF	Estimate	Standard Error	Wald 95% Confidence Limits		Wald Chi-Square	Pr > ChiSq
				Lower	Upper		
Intercept	1	-0.5461	0.1650	-0.8696	-0.2227	10.95	0.0009

```
TITLE1 "ZINB Model of Aggressive Verbalizations";
PROC GENMOD DATA=example NAMELEN=100;
  CLASS ERconds; * CLASS statement creates contrasts for us;
  MODEL Aggr = Neutral IPV ERconds IPV*ERconds / LINK=LOG DIST=ZINB;
  ZEROMODEL / LINK=LOGIT; RUN;
```

Criteria For Assessing Goodness Of Fit	
Criterion	Value
Full Log Likelihood	-408.9731
AIC (smaller is better)	835.9463
AICC (smaller is better)	836.7835
BIC (smaller is better)	866.6912

Negative Binomial $-2LL = 817.95$,
 AIC =833.95, BIC = 861.28

AIC and BIC are higher for ZINB, so NB is enough.
 Below, the logit of being an “extra 0” = $-22!$

Analysis Of Maximum Likelihood Zero Inflation Parameter Estimates

Parameter	DF	Estimate	Standard Error	Wald 95% Confidence Limits		Wald Chi-Square	Pr > ChiSq
				Lower	Upper		
Intercept	1	-22.2405	39404.29	-77253.2	77208.75	0.00	0.9995